COMPASS for Pressure

Test macro to Zero Fluke 700P and 750P pressure modules with COMPASS for Macro with 700PCK v4

```
'This macro zero's the Fluke 700P pressure modules
'************************************************

Function Fluke700P_Zero(iT, iL, iC, iP, cTest, cConfig)

    If iC > 1 Then
        'Only zero on first cycle
        Exit Function
    End If

    If cCOMPASS.SystemAbort = True Or cCOMPASS.cConfig.DUTPrs(1).GetParamData(1) = 1 Then
        cCOMPASS.AbortTest True
        Exit Function
    End If

    cDebug.LogStatus "Preparing to run Fluke 700P_Runzero"
    Fluke700P_RunZero
    cDebug.LogStatus "Completed run Fluke 700P_Runzero"

End Function


Function Fluke700P_RunZero()
    'This macro stores the active pressure and subtracts it from
    'each reading. No internal changes to the module are made
    cCOMPASS.StatusDisplay "Adjusting module zero..."
    cDebug.LogStatus "Adjusting Module Zero"

    'Set the zero value based on the
    MinP = cCOMPASS.cConfig.DUTPrs(1).RangeMain.MinFinal
    If cCOMPASS.cConfig.DUTPrs(1).RangeMain.measmode = 1 And MinP =0 Then
      ' Gauge pressure measurement mode
      'Vent the pressure
      cDebug.LogStatus "Venting system Pressure"
      cCompass.cConfig.RefPrs(1).ioSetOutput 0,1,0
      PVent = True

    Else
```

```
  'Set the target of the min pressure value, whatever it may be
  RUnit = cCompass.cConfig.RefPrs(1).RangeMain.UnitFinal
  DUnit = cCompass.cConfig.DUTPrs(1).RangeMain.UnitFinal
  PTarget = cCOMPASS.UnitConversion(CDbl(MinP),CInt(RUnit),CInt(DUnit),0)
  cCompass.cConfig.RefPrs(1).ioSetOutput CDbl(PTarget), 0,0
  PVent = False

End If


TimeDelay 120

If Fluke700P_WaitForReady(1, 120) = False Then Exit Function
If cCOMPASS.SystemAbort Then Exit Function
  cDebug.LogStatus "Wait for Ready Complete"

Fluke700P_Dwell 10
If cCOMPASS.SystemAbort Then Exit Function
cDebug.LogStatus "Dwell timeComplete"

''' this assumes the interface macro has executed and the dut is initialized.
For i = 1 To cCOMPASS.cConfig.DUTPrs.Count
    Manuf = cCOMPASS.cConfig.DUTPrs(i).RangeMain.GetParent.Manufacturer
    sn = cCOMPASS.cConfig.DUTPrs(i).RangeMain.GetParent.sn
    cDebug.LogStatus "Zero DUT: " & i & ") Manufacturer=" & Manuf & ",SN=" & sn

    If InStr(Ucase(Manuf), "FLUKE") Then
        cCOMPASS.cConfig.DUTPrs(i).SetParamData 0, 0 'Remove existing zero correction
        cCOMPASS.cConfig.DUTPrs(i).RangeMain.FinalOutput = -9999

        'read and average pressure
        DPrs = Fluke700P_ReadPressure(5, i)
        RPrs = cCOMPASS.cConfig.RefPrs(1).GetParamData(1)
        cDebug.LogStatus i & ") DPrs=" & DPrs & ",RPrs=" & RPrs
        If cCOMPASS.SystemAbort Then Exit Function

            If PVent = True Then
                'When the pressure is vented, the reference value is assumed to be 0
                ZeroPrs = DPrs
            Else
                ZeroPrs = DPrs - RPrs

            End If
```

```
            'Store offset in data file as Calibration Coefficient #1.
            'The offset is available in Reports.
            dUnitText = cConfig.DUTPrs(i).RangeMain.UnitFinalText
            cDebug.LogStatus i & ") PZero:" & ZeroPrs & " " & dUnitText
            cConfig.DUTPrs(i).SetParamData 0, CDbl(ZeroPrs)

            cConfig.DUTPrs(i).RangeMain.GetParent.CalInfo1 =FormatNumber(ZeroPrs, 4)

            cDebug.LogStatus "Set the CalInfo"

            If (cCOMPASS.COMPASSRunState And 2^6) Then
                'Data file exists so OK to write to the object
                cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient1 = FormatNumber(ZeroPrs, 4)
            End If

        End If
    Next

    cCOMPASS.StatusDisplay ""

End Function


Function Fluke700P_Dwell(dwell)
    cDebug.LogStatus "Dwell for " & dwell & "s"
    tStart = Timer
    Do
        cCOMPASS.TimeDelay 1
        td = CInt(dwell - Time_Difference(tStart))
        cCOMPASS.StatusDisplay "Dwell..." & td
        If cCOMPASS.SystemAbort Then Exit Function
        If td <= 0 Then Exit Do
    Loop Until False
End Function

Function Fluke700P_WaitForReady(isZero, timeout)
    cDebug.LogStatus "Wait for Ready: timeout=" & timeout
    tStart = Timer
    Do
        cCOMPASS.TimeDelay 2    'delay controller
        cCOMPASS.StatusDisplay process & ": Waiting for Ready........."
```

```
        If cCOMPASS.SystemAbort Then Exit Function

        If Time_Difference(tStart) > timeout Then
            cDebug.LogStatus "*******TIMEOUT WAITING for Ready****"
            If (cCOMPASS.cConfig.DUTPrs(1).RangeMain.MeasMode = 0) And (isZero = 1) Then
                cDebug.LogStatus "Timeout for absolute 0"
                Exit Do 'absolute 0, as good as possible
            End If
            Fluke700P_SetPressure = False
            Exit Function
        End If

        If cConfig.SetPrs(1).RangeMain.Useready Then
            cDebug.LogStatus "Wait for ready...SetPrs.Ready=" & cConfig.SetPrs(1).Ready
            If cConfig.SetPrs(1).Ready Then Exit Do
        Else
            cDebug.LogStatus "Wait for ready...RefPrs.Ready=" & cConfig.RefPrs(1).Ready
            If cConfig.RefPrs(1).Ready Then Exit Do
        End If
    Loop
    Fluke700P_WaitForReady = True
End Function

Function Fluke700P_ReadPressure(nAvg, iDUT)
    DPrs = -9999
    DSum = 0
    DCnt = 0
    RPrs = -9999
    RSum = 0
    RCnt = 0

    cCOMPASS.StatusDisplay "Averaging DUT Pressure..."
    cDebug.LogStatus iDUT & ") Averaging DUT Pressure"

    Do
        cCOMPASS.TimeDelay 1
        If cCOMPASS.SystemAbort Then Exit Function

        DPrs = cConfig.DUTPrs(iDUT).RangeMain.FinalOutput
        If DPrs <> -9999 Then
            DSum = DSum + DPrs   'Update Sum
            DCnt = DCnt + 1      'Update Increment
        End If
```

```
        RPrs = cConfig.RefPrs(1).RangeMain.FinalOutput
        If RPrs <> -9999 Then
            RSum = RSum + RPrs   'Update Sum
            RCnt = RCnt + 1      'Update Increment
        End If

    Loop While DCnt < nAvg And RCnt < nAvg


    cDebug.LogStatus iDUT & ") DUT Sum:" & DSum
    cDebug.LogStatus iDUT & ") DUT Count:" & DCnt
    DPrs = DSum / DCnt
    cDebug.LogStatus iDUT & ") DUT Avg:" & DPrs

    cDebug.LogStatus iDUT & ") Ref Sum:" & RSum
    cDebug.LogStatus iDUT & ") Ref Count:" & RCnt
    RPrs = RSum / RCnt
    cDebug.LogStatus iDUT & ") Ref Avg:" & RPrs

    'Reference pressure is averaged in it's Unit of measure
    'Convert to DUT's Unit of Measure
    dUnit = cConfig.DUTPrs(iDUT).RangeMain.UnitFinal
    rUnit = cConfig.RefPrs(1).RangeMain.UnitFinal
    PRef = cCOMPASS.UnitConversion(CDbl(RPrs), CInt(dUnit), CInt(rUnit), 0)
    cDebug.LogStatus iDUT & ") Ref Avg DUT Unit:" & PRef

    'store Ref prs
    cCOMPASS.cConfig.RefPrs(1).SetParamData 1, CDbl(PRef)
    'return DUT prs
    Fluke700P_ReadPressure = DPrs
End Function
```